> restart; $t0 := time()$:

> $H0 := \text{binomial}(n, k)\hat{}7;$

$$H0 := \binom{n}{k}^7 \quad (1)$$

> $H := H0 / (2*n + 3*k);$

$$H := \frac{\binom{n}{k}^7}{2\,n + 3\,k} \quad (2)$$

The telescoper L in Q(n)[Sn] of H is very large. Instead of computing L all at once,
lets try to find factors of L, one at a time.

Factors correspond to submodules or quotient modules. Lets try to find some.

Finding a natural submodule N of M.

Let Omega = Q(n, k) * H0. This is a Q(n,k)[Sn, Sn^(-1), Sk, Sk^(-1)]-module.

Let M = Omega / Delta_k(Omega). This is a Q(n)[Sn, Sn^(-1)]-module.

Goal: compute the telescoper of H, which is the minimal annihilator L for the image of H in M.

If you apply Sn or Sk to an element of Omega, then H0 gets multiplied by:

> $R1 := simplify\left(convert\left(\frac{subs(n = n + 1, H0)}{H0}, \text{GAMMA}\right)\right);$

$$R1 := -\frac{(n+1)^7}{(-n+k-1)^7} \quad (1.1)$$

> $R2 := simplify\left(convert\left(\frac{subs(k = k + 1, H0)}{H0}, \text{GAMMA}\right)\right);$

$$R2 := -\frac{(-n+k)^7}{(k+1)^7} \quad (1.2)$$

Let N be the image of Q(n)[k][Sn, Sn^(-1)] * H0 in M.

This is a natural submodule of M that does NOT contain H. To see this,
note that Sn, Sn^(-1), Sk, Sk^(-1) can only introduce denominators of the form

n + integer, k + integer, k-n + integer

but not 2n + 3k + integer.

The minimal annihilator of H in M/N is a right-factor R the telescoper L.

Computing the annihilator R of the image of H in M/N.

If R(H) is zero in M/N, then R(H) is in N, which means the denominator 2n + 3k seen in H is gone.
To get rid of this denominator, we need to cancel it against something with the same denominator.

```
> H_shift := subs(n = n + 3, k = k-2, H);
```

$$H\_shift := \frac{\binom{n+3}{k-2}^7}{2\,n+3\,k} \qquad (2.1)$$

H_shift and Sn^3(H) have the same image in M because k-shifts act trivially on M.

Next, we want to find r in Q(n) such that r*H has the same residue as H_shift
(so that their denominators cancel).

```
> RatFunction := simplify(convert(H_shift / H, GAMMA)) :
  r := factor(subs(k = -2/3 * n, RatFunction));   # quotient of residues
```

$$r := \frac{1338925209984\,(n+2)^7\,(n+1)^7\,n^7\,(2\,n+3)^7}{78125\,(5\,n+12)^7\,(5\,n+9)^7\,(5\,n+6)^7\,(5\,n+3)^7} \qquad (2.2)$$

```
> R := Sn^3 - r;                          # should cancel out 2·n+3·h
```

$$R := Sn^3 - \frac{1338925209984\,(n+2)^7\,(n+1)^7\,n^7\,(2\,n+3)^7}{78125\,(5\,n+12)^7\,(5\,n+9)^7\,(5\,n+6)^7\,(5\,n+3)^7} \qquad (2.3)$$

```
> R := collect(primpart(R, Sn), Sn, factor);        # make R fraction-free.
```

$$R := 78125\,Sn^3\,(5\,n+12)^7\,(5\,n+9)^7\,(5\,n+6)^7\,(5\,n+3)^7 - 1338925209984\,(n+2)^7\,(n+1)^7\,n^7\,(2\,n+3)^7 \qquad (2.4)$$

Even though the telescoper L of H is very large, we found an order-3
right-factor of L with practically zero CPU time!

The corresponding left-factor is the telescoper of R(H).
Lets compute this next.
R annihilates H in M/N, therefore, R(H) is in N.

A basis of N.

N is the image of Q(n)[k][Sn, Sn^(-1)] * H0 in M.
Reducing modulo Delta_k(Omega), standard procedure in telescoping algorithms,
will simplify every element of N to this form:

R * H0

with R in Q(n)[k] and degree(R, k) <= 6. So this is a Q(n)-basis of N:

Basis = {1, k, k^2, k^3, k^4, k^5, k^6}   (where we omitted the factor H0).

Conclusion: N is a Q(n)[Sn]-module of dimension 7.

So any element of N has an annihilator of order <= 7.

L = the annihilator of R(H) times R.

Reducing R(H) to express it in terms of a basis of N.

```
> RH := lcoeff(R, Sn) * subs(n = n + 3, k = k-2, H) + tcoeff(R, Sn) * H;
```

$$RH := \frac{78125\,(5\,n+12)^7\,(5\,n+9)^7\,(5\,n+6)^7\,(5\,n+3)^7\binom{n+3}{k-2}^7}{2\,n+3\,k} - \frac{1338925209984\,(n+2)^7\,(n+1)^7\,n^7\,(2\,n+3)^7\binom{n}{k}^7}{2\,n+3\,k} \qquad (4.1)$$

In order to represent R(H) with a rational function, we divide RH by H0:

```
> RHd := normal(simplify(convert(subs(k = n-k, RH/H0), GAMMA))) :
  # subs(k = n-k ..) makes it easier to code the reduction:

  for j from 4 to 0 by -1 do
      G := add(c[i]*k^i, i = 0..6) / (k + j)^`if`(j = 0, 0, 7);
      G := subs(k = k + 1, G) * ((n-k) / (k + 1))^7 - G;
      eq := {coeffs(rem(numer(normal( (RHd - G) * (k + j + 1)^7 )), (k + j + 1)^7, k),
     k)};
      RHd := normal(RHd - subs(solve(eq, {seq(c[i], i = 0..6)}), G));
   od:
```

Above we applied an ad-hoc reduction of R(H) modulo Delta_k(Omega),
to write R(H) as a Q(n)-linear combination of {1, k, k^2, k^3, k^4, k^5, k^6} (times H0).

We'll actually use a slightly different basis, the reason will be explained in the next section.

```
> BasisN := [1, u, u^2, u^3, v, v·u, v·u^2];
```

$$BasisN := [1, u, u^2, u^3, v, v\,u, v\,u^2] \qquad (4.2)$$

where

```
> u := k*(n-k);      # Invariant under phi  (more details in the next section)
  v := k - (n-k);    # Anti-invariant under phi
```

$$u := k\,(n-k)$$
$$v := 2\,k-n \qquad (4.3)$$

```
> {coeffs(collect(RHd - add(c[i]*u^i, i = 0..3) + v*add(d[i]*u^i, i = 0..2), k), k)} :
  Decomp := factor(solve(%, {seq(c[i], i = 0..3), seq(d[i], i = 0..2)})) :
```

This computation wrote R(H) as a linear combination of BasisN (omitting the factor H0).

Using automorphisms to construct submodules.

The Zeilberger program in Maple takes 31.5 seconds to compute the telescoper L of H.

It has order 10. That is not surprising because R has order 3, and R(H) is in N,
which is a module of dimension 7.
We computed this order-3 right-factor R of L in about 0.01 seconds,
a **tiny** fraction of the time it takes to compute the full telescopers.

The idea was to compute in M/N instead of in M.
Lets try something similar for computing the telescoper of R(H),
the left-factor of L that we still have to find.

Let
phi: N --> N send k to n-k.

This is an **automorphism** of N because phi(H0) = H0.
It has order 2, so it has eigenvalues +1 and -1.
Let N+ be the eigenspace for +1, and N- be the eigenspace for -1.

If u = k * (n-k) then phi(u) = u. So a basis for N+ is: 1, u, u^2, ...
If v = k - (n-k) then phi(v) = -v. So a basis of N- is: v, v * u, v^2 * u, ...

In a previous section, we wrote R(H) as a linear combination of the basis elements of N+ and N-.
This gives us the projections of R(H) on N+ and on N-.

Let L+ be the annihilator of the projection of R(H) on N+.
Let L- be the annihilator for the projection of R(H) on N-.

To compute L+ we first compute the action of Sn on the basis of N+.
Then we get L+ via a cyclic vector computation.

The action of Sn on a basis of N+

Here we combine the basis elements in B+ by taking a linear combination with variables c[i] as weights.
This way we can apply Sn to all elements of B+ at once.

```
> BP := add(c[i] * u^i, i = 0..3);
```

$$BP := c_0 + c_1 k (n-k) + c_2 k^2 (n-k)^2 + c_3 k^3 (n-k)^3 \quad \textbf{(6.1)}$$

Apply Sn to "basis" BP:

```
> SnBP := subs(n = n + 1, BP) * ((n + 1)/(n-k + 1))^7 :
  SnBP := subs(k = n - k, SnBP) :
```

Take a generic element of Delta_k(Omega)   (with the factor H0 removed)

```
> G := add(e[i]*k^i, i = 0..6) :
```

```
G := subs(k=k+1, G)*((n-k)/(k+1))^7 - G:
```

Now reduce modulo G; compute the unknown coefficients in G.

```
> sol := solve({coeffs(rem(normal((SnBP - G)*(k+1)^7), (k+1)^7, k), k)}, indets(G)
      minus {k, n}):
  SnBP := normal(SnBP - subs(sol, G)):
```

Rewrite in terms of the basis 1, u, u^2, ... of N+ instead of the basis 1, k, k^2, ... of N.
Then we can read off the matrix M.

```
> SnBP := evala(subs(k=RootOf(u - U, k), SnBP)): # Write SnBP in terms of u instead of k.
  M := Matrix([seq([seq(factor(coeff(coeff(SnBP, c[i]), U, j)), i=0..3)], j=0..3)]);
```

$$M := \left[\left[\frac{1717 n^6 + 1293 n^5 + 730 n^4 + 306 n^3 + 93 n^2 + 19 n + 2}{(n+1)^6}, \frac{(462 n^5 + 330 n^4 + 165 n^3 + 55 n^2 + 11 n + 1) n}{(n+1)^4}, \frac{(126 n^4 + 84 n^3 + 36 n^2 + 9 n + 1) n^2}{(n+1)^2}, (35 n^3 + 21 n^2 + 7 n + 1) n^3\right], \left[-\frac{14 (643 n^4 + 355 n^3 + 138 n^2 + 34 n + 4)}{(n+1)^6}, -\frac{2441 n^4 + 1315 n^3 + 485 n^2 + 111 n + 12}{(n+1)^4}, -\frac{672 n^4 + 348 n^3 + 117 n^2 + 23 n + 2}{(n+1)^2}, -(189 n^3 + 91 n^2 + 25 n + 3) n\right], \left[\frac{42 (263 n^2 + 75 n + 10)}{(n+1)^6}, \frac{14 (215 n^2 + 61 n + 8)}{(n+1)^4}, \frac{5 (167 n^2 + 47 n + 6)}{(n+1)^2}, 238 n^2 + 66 n + 8\right], \left[-\frac{1848}{(n+1)^6}, -\frac{504}{(n+1)^4}, -\frac{140}{(n+1)^2}, -40\right]\right] \quad \textbf{(6.2)}$$

M gives the action of Sn on the basis B+

Computing L+ with a cyclic vector computation using matrix M.

The projection of R(H) on N+ written in terms of basis B+ is given by:

```
> V[0] := [seq(subs(Decomp, c[i]), i=0..3)]:
```

Use matrix M (the action of Sn on B+) to apply Sn four times:

```
> for i to 4 do
        V[i] := map(factor, convert(M . Vector(subs(n = n + 1, V[i-1])), list))
  od:
```

A linear relation between V[0] .. V[4] gives L_plus:

```
> L_plus := subs(solve({op(add(c[i]*~ V[i], i = 0 ..4))}, {seq(c[i], i = 0 ..4)}), add(c[i]* Sn
        ^i, i = 0 ..4)) :
  L_plus := collect(primpart(L_plus, Sn), Sn) : # Large expression, use ; instead of : to view it
```

The same computation for L-

```
> BM := v * add(d[i] * u^i, i = 0 ..2) : # Basis for N-
  # Applying Sn:
  SnBM := subs(n = n + 1, BM) * ((n + 1) / (n-k + 1))^7 :
  SnBM := subs(k = n-k, SnBM) :

  sol := solve({coeffs(rem(normal( (SnBM - G) * (k + 1)^7 ), (k + 1)^7, k), k)},
        indets(G) minus {k, n}) :
  SnBM := normal(SnBM - subs(sol, G)) : # Reduction mod Delta_k( Omega ).
  SnBM := evala(subs(k = RootOf( u - U, k), -SnBM / v)) : # Write SnBM in terms of B-
  M := Matrix( [seq([seq(factor(coeff(coeff(SnBM, d[i]), U, j)), i = 0 ..2)], j = 0 ..2)] );
```

$$M := \left[\left[-\frac{131\,n^4+160\,n^3+100\,n^2+34\,n+5}{(n+1)^4},\ -\frac{42\,n^4+48\,n^3+27\,n^2+8\,n+1}{(n+1)^2},\ -\left(14\,n^3+14\,n^2+6\,n+1\right)n\right],\ \left[\frac{14\left(17\,n^2+10\,n+2\right)}{(n+1)^4},\ \frac{79\,n^2+46\,n+9}{(n+1)^2},\ 28\,n^2+16\,n+3\right],\ \left[-\frac{42}{(n+1)^4},\ -\frac{14}{(n+1)^2},\ -5\right]\right] \qquad \textbf{(8.1)}$$

This matrix gives the action of Sn on the basis B-

Use it to compute L-, the annihilator of the projection of R(H) on N-.

```
> V[0] := [seq(subs(Decomp, d[i]), i = 0 ..2)] :
        # Projection of R(H) on N-  written in terms of basis B-
  for i to 3 do
        V[i] := map(factor, convert(M . Vector(subs(n = n + 1, V[i-1])), list))
  od:
  L_minus := subs(solve({op(add(d[i]*~ V[i], i = 0 ..3))}, {seq(d[i], i = 0 ..3)}), add(d[i]
        * Sn^i, i = 0 ..3)) :
  L_minus := collect(primpart(L_minus, Sn), Sn) :
```

The complete telescoper for H.

We started by computing a right-factor R of the telescoper.
This R was the minimal operator that can remove the 2n+3k denominator,
i.e. the minimal operator for which R(H) is in N.

The corresponding left-factor is the telescoper of R(H).

Because we found an automorphism, we could decompose N as a direct sum of two submodules, N+ and N-.
Annihilating R(H) is equivalent to annihilating both of its components.
The annihilators of these components were L+ and L-

Hence: The telescoper of R(H) is LCLM(L+, L-).

and: The telescoper of H is LCLM(L+, L-) times R.

This telescoper is of the form: L = LCLM(order4, order3) times order3.

We computed these factors R, L+, and L- in this amount of time:

```
> time( ) - t0;
                                1.874                                (9.1)
```

Which is **many times faster** than Maple's Zeilberger algorithm takes to compute L.
Moreover, the factored form is **more useful** since it is much smaller in size.

Elements of N- are anti-symmetric and contribute 0 to the sequence sum(H, k = 0..n) (n=1,2,...).

So L- contributes 0 to the sequence.
Hence: L+ times R will also annihilate the sequence.
It has order 7 and is the minimal recurrence.

Exercises

Let H0 = binomial(n, k)^s.
Let Omega = Q(n,k) * H0.

Let M = Omega / Delta_k(Omega).
Let N = image of Q(n)[Sn, Sn^(-1)] * H0 in M.

Let r = floor((s+1)/2).

(1) Show that Delta_k(Omega) contains polynomials with k-degrees 2*r - 1, 2*r, 2*r + 1, ...

(2) Show that { 1, k, k^2, ..., k^(2*r - 2) } (times H0) is a basis of N,
so dim(N) = 2*r - 1.

(3) Show that { 1, u, u^2, ..., u^(r-1) } (times H0) is a basis of N+,

so dim(N+) = r.

(4) Show that the telescoper of H0 has order at most r.
(Theorem 1.1 in [Straub, Zudilin] says that the order is at least r)

(5) Show that { v, u*v, ..., u^(r-2)*v } (times H0) is a basis of N-,
so dim(N-) = r-1.

(6) Show that the telescoper of v * H0 has order at most r-1,
where v = 2*k - n.

Research questions

The characteristic polynomials of L+ and L- are:

```
> factor(primpart(lcoeff(L_plus, n))); factor(primpart(lcoeff(L_minus, n)));
```

$$(Sn - 128)\,(Sn^3 + 57\,Sn^2 - 289\,Sn - 1)$$

$$Sn^3 + 57\,Sn^2 - 289\,Sn - 1 \qquad \textbf{(11.1)}$$

The roots of the characteristic polynomial of Telescoper(binomial(n,k)^s) are
{ (z + z^(-1))^s | z^s = 1, z ≠ -1, z^2 ≠ -1 }, where the root 2^s appears only in L+ but not in L-.

(1): If L is the telescoper of H, how to compute invariant data (like the characteristic polynomial, or the p-curvature) directly from H, without computing L?

(2): Apart from denominators or automorphisms, what other ways can we find submodules?

(3): Let M0 consist of those elements h in M for which sum_k(h) = 0 for all n >> 0.
In our example, N- is a submodule of M0.
But in general, how do we decide if M0 is {0} or not? How do we find elements?

Let L_min := **MinimalRecurrence**(sum_k(h)). If L ≠ L_min then we found a non-zero element L_min(h) in M0, but found it too late to expedite the computation of L.

(4): How to best implement submodules for hypergeometric/hyperexponential/D-finite telescoping?

```
>
```